
Wish list for computational imaging code J. Fessler
CI in Julia

▶ Performance of a compiled language

JIT via LLVM

▶ Interactive development / prototyping

Dynamic typing / Jupyter notebooks

▶ Readability
◦ Syntax matching the mathematics of computational imaging

Unicode / UTF-8

◦ Built-in operations with numerical arrays

By design

◦ Namespace control

using , import

▶ Relevant libraries (FFT, linear algebra, statistics, autograd, ...)

FFTW, LAPACK, ...

▶ Easy use of distributed computing, GPU resources

Yes

▶ Library ecosystem that facilitates reproducibility

git, Pkg, Manifest

▶ Free / open source

Yes

▶ IDE / debugger

VSCode, ...

▶ Code reuse (e.g., “object oriented”)

multiple dispatch, subtypes

▶ Memory efficient

Float16, Sparse of any type, call by ref.

▶ Interoperable with other languages

ccall, pycall, ...

Julia: incomplete history J. Fessler
CI in Julia

▶ https://julialang.org
▶ Started in 2009 around MIT (Alan Edelman, Jeff Bezanson, Stefan Karpinski, and Viral B. Shah)

collaboration between computer sciences and computational sciences
▶ First release in 2012
▶ Version 1.0 in Aug. 2018
▶ 2017 SIAM Review paper [1]
▶ 2017 Forbes magazine article
▶ 2019 InfoWorld comparison of Julia and Python
▶ 2019 Nature article about Julia
▶ 2019 SIAM CSE Conference: James H. Wilkinson Prize for Numerical Software
▶ Sponsors of Juliacon 2018 and Juliacon 2019 include:

Microsoft, Amazon, Google, Intel, NVIDIA, CapitalOne, JP Morgan, ...
▶ Best of Matlab, Python, LISP, ... Addresses “2 (or 3?) language problem”

4 / 36

https://julialang.org
https://www.forbes.com/sites/suparnadutt/2017/09/20/this-startup-created-a-new-programming-language-now-used-by-the-worlds-biggest-companies/#9af992d7de2a
https://www.infoworld.com/article/3241107/python/julia-vs-python-julia-language-rises-for-data-science.html
http://doi.org/10.1038/d41586-019-02310-3
https://sinews.siam.org/Details-Page/scientific-machine-learning-how-julia-employs-differentiable-programming-to-do-it-best
http://juliacon.org/2018/
http://juliacon.org/2019/


UM - history with Julia J. Fessler
CI in Julia

▶ Used at UM since 2017 for teaching
▶ EECS 551: Matrix methods in signal processing and machine learning
▶ EECS 505: Computational data science and machine learning
▶ EECS 556: Image processing
▶ EECS 559: Optimization methods for signal and image processing
▶ ROB 101: Computational linear algebra

▶ Michigan Image Reconstruction Toolbox (MIRT)
▶ Matlab version (inactive) https://github.com/JeffFessler/mirt
▶ Julia version (active) https://github.com/JeffFessler/MIRT.jl

▶ Group now uses mix of Julia, Matlab, python...

▶ Growing use elsewhere, e.g.,
ENGR108 at Stanford (Intro to Matrix Methods)

https://web.eecs.umich.edu/~fessler/course/551
https://ece.engin.umich.edu/academics/course-information/course-descriptions/eecs-505
https://web.eecs.umich.edu/~fessler/course/556
https://web.eecs.umich.edu/~fessler/course/598
https://robotics.umich.edu/academic-program/course-offerings/rob101/
https://github.com/JeffFessler/mirt
https://github.com/JeffFessler/MIRT.jl
https://web.stanford.edu/class/engr108


Julia-based book J. Fessler
CI in Julia

• All examples in Julia
• Online demos:

https://github.com/JeffFessler/
book-la-demo

• Topics include: low-rank matrix approximation,
robust PCA, photometric stereo, matrix comple-
tion, ...

• Available in Europe/UK now;
in US: July 2024 from Cambridge Univ. Press

https://github.com/JeffFessler/book-la-demo
https://github.com/JeffFessler/book-la-demo
https://www.cambridge.org/highereducation/isbn/9781009418140


Syntax comparison J. Fessler
CI in Julia

Operation Matlab Julia Python import numpy as np

Dot product dot(x,y) dot(x,y) np.dot(x,y)
Matrix mult. A * B A * B A @ B
Element-wise A .* B A .* B A * B
Scaling 3 * A 3A or 3*A 3 * A
Matrix power A^2 A^2 np.linalg.matrix_power(A,2)
Element-wise A.^2 A.^2 A**2
Inverse inv(A) inv(A) np.linalg.inv(A)
Inverse A^(-1) A^(-1) np.linalg.inv(A)
Indexing A(i,j) A[i,j] A[i,j]
Range 1:9 1:9 np.arange(1,9,1)
Range linspace(0,4,9) range(0,4,9) np.arange(0,4.01,0.5)
Strings ’text’ "text" (either)
Inline func. f = @(x,y) x+y f(x,y) = x+y f = lambda x,y : x+y
Increment A = A + B A += B A += B
Herm. transp. A’ A’ A.conj().T[
1 2
3 4

]
[1 2; 3 4] [1 2; 3 4] np.array([[1, 2], [3, 4]])

See https://cheatsheets.quantecon.org for more. 7 / 36

https://cheatsheets.quantecon.org


Function definitions J. Fessler
CI in Julia

Ways to define the function f (x , y) = x2 + y3 in Julia:

▶ f(x,y) = x^2 + y^3

▶ f = (x,y) -> x^2 + y^3

▶ function f(x,y)
return x^2 + y^3

end

▶ function f(x,y)
x^2 + y^3 (Functions return the last value evaluated.)

end

Kronecker sum example (skip) J. Fessler
CI in Julia

▶ Kronecker sum definition in math, for M × M matrix A: and N × N matrix B:

A ⊕ B ≜ (IN ⊗ A) + (B ⊗ IM).

▶ Kronecker sum function in Julia:� �
using LinearAlgebra: I
⊗(A,B) = kron(A,B) # Kronecker product
⊕(A,B) = I(size(B,1)) ⊗ A + B ⊗ I(size(A)[1]) # Kronecker sum
X = ones(2,2) ⊕ [1 2; 3 4] # test drive� �

▶ using ... controls name-space

▶ Enter symbols like ⊗ with LATEXcodes and tab completion:
� �
\otimes<tab>� �

▶ Julia supports argument chaining, e.g., size(A)[1]

9 / 36



Reproducibility example J. Fessler
CI in Julia

Trotier et al., “Accelerated 3D multi-echo spin-echo sequence with a subspace
constrained reconstruction for whole mouse brain T2 mapping” MRM 2024 [2]
https://github.com/CRMSB/PAPER_subspace_MESE/blob/main/Project.toml
name = " Subspace_MESE "
uuid = "03 ee4b65 -1735 -4169 -82da -5 df6ad43f66c "
authors = [" aTrotier <a. trotier@gmail .com > and contributors "]
version = " 1.0.2 "

[deps]
LinearOperatorCollection = "a4a2c56f -fead -462a-a3ab -85921 a5f2575 "
BartIO = "8b90e6a2 -6e4d -4906 -89ae - b5f72ca4a65b "
EPGsim = "b6a82cc1 -40c9 -4006 -90b7 -9176002 c0410 "
FFTW = "7a1cc6ca -52ef -59f5 -83cd -3 a7055c09341 "
LinearAlgebra = "37 e2e46d -f89d -539d-b4ee -838 fcccc9c8e "
LsqFit = "2fda8390 -95c7 -5789 -9 bda -21331 edee243 "
MRICoilSensitivities = "c57eb701 -aafc -44a2 -a53c -128049758959 "
MRIFiles = "5a6f062f -bf45 -497d-b654 - ad17aae2a530 "
MRIReco = "bdf86e05 -2d2b -5731 - a332 - f3fe1f9e047f "

[ compat ]
julia = "1.9"
LinearOperatorCollection = " =1.1.2 "
BartIO = " 0.3.1 "
EPGsim = " 0.1.1 "
FFTW = " 1.7.2 "
LsqFit = " 0.15.0 "
MRICoilSensitivities = " 0.1.3 "
MRIFiles = " 0.2.0 "
MRIReco = " 0.8.1 "

Gradient descent example J. Fessler
CI in Julia

To minimize f (x) using gradient descent (for illustration) in math:

xk = xk−1 − α∇f (xk−1), for k = 1, 2, . . . , Niter.

Gradient descent function code in Julia looks remarkably similar:� �
function gd(∇f::Function, α::Real, x ; Niter::Int = 50)

for iter in 1:Niter
x = x - α * ∇f(x)

end
return x

end� �
Enter symbols like α with LATEXcodes and tab completion:

� �
\alpha<tab>� �

11 / 36



Julia types and function arguments J. Fessler
CI in Julia� �

function gd(∇f::Function, α::Real, x ; Niter::Int = 50)� �
▶ Type annotation (via :: ) is optional (cf. x )
▶ Functions are first-class object types in Julia
▶ Julia is dynamically typed
▶ Real is an “abstract supertype” (step size α is any type of real number)
▶ subtypes(Real) returns

[AbstractFloat,AbstractIrrational,Integer,Rational]

▶ subtypes(AbstractFloat) returns [Float16,Float32,Float64,BigFloat]
(concrete number types: stored as bits)

▶ Optional named arguments (after ; ) with default values

12 / 36



Test GD routine (skip) J. Fessler
CI in Julia

f (x) = 1
2 ∥Ax − y∥2

2 =⇒ ∇f (x) = A′(Ax − y), L∇f = ∥A∥2
2 , α = 1/L∇f� �

using LinearAlgebra: opnorm # same as svdvals(A)[1]
include("gd.jl") # provides gd()
M,N = 6,4
A = randn(M,N); y = randn(M); # test data
∇f(x) = A'*(A*x - y) # LS gradient
xh = A \ y # global minimizer of f
α = 1/opnorm(A)ˆ2 # step size
xgd = gd(∇f, α, zeros(N) ; Niter=9000)
@assert xgd ≈ xh # equivalent within precision of type� �

▶ a ≈ b equivalent to isapprox(a,b)

Regularized LS solver (e.g., for computational imaging) J. Fessler
CI in Julia

x̂ = arg min
x

Ψ(x), Ψ(x) ≜ 1
2 ∥Ax − y∥2

2 + R(x)

=⇒ ∇Ψ(x) = A′(Ax − y) + ∇R(x)� �
include("gd.jl") # provides gd() function
function rls(y, A, x0, α::Real, ∇R::Function ; Niter::Int = 50)

∇Ψ(x) = A'*(A*x - y) + ∇R(x) # reg. LS cost gradient
gd(∇Ψ, α, x0 ; Niter) # returns final x

end� �
▶ Could add type annotations like y::AbstractVector and A::AbstractArray

or A::Matrix{Float32}
▶ Using “duck typing” can facilitate code reuse

14 / 36



Regularized LS test (skip) J. Fessler
CI in Julia

R(x) = β
1
2 ∥x∥2

2 =⇒ x̂ = arg min
x

Ψ(x) = (A′A + βI)−1A′y� �
using LinearAlgebra: I, opnorm # same as svdvals(A)[1]
include("rls.jl")
M,N = 6,4
A = randn(M,N); y = randn(M); # test data
β = 2; ∇R(x) = β*x # Tikhonov regularizer gradient
xh = (A'A + β*I) \ (A'*y) # global minimizer of Ψ
α = 1/(opnorm(A)ˆ2 + β) # step size
xr = rls(y, A, zeros(N), α, ∇R ; Niter=200)
@assert xr ≈ xh # equivalent within precision of type� �

Linear operators and overloading J. Fessler
CI in Julia

▶ Develop / debug code with A as a (small, dense) matrix
▶ For most computational imaging, A too big to store explicitly
▶ Want to avoid rewriting new versions of GD and RLS for every A
▶ Apply code with more general linear operator A that can do A*x and A’*y
▶ Minimalist version of “operator overloading:”� �

struct LinOp # see LinearMaps.jl and LinearMapsAA.jl for pro version
forward::Function # A * x will call forward(x)
adjoint::Function # A' * y will call adjoint(y)

end

Base.:(*)(A::LinOp, x) = A.forward(x) # for A*x
Base.adjoint(A::LinOp) = LinOp(A.adjoint, A.forward) # for A'� �

▶ Full-scale version: https://github.com/JeffFessler/LinearMapsAA.jl

Multiple dispatch introduction J. Fessler
CI in Julia� �

struct LinOp # see LinearMaps.jl and LinearMapsAA.jl for pro version
forward::Function # A * x will call forward(x)
adjoint::Function # A' * y will call adjoint(y)

end

Base.:(*)(A::LinOp, x) = A.forward(x) # for A*x
Base.adjoint(A::LinOp) = LinOp(A.adjoint, A.forward) # for A'� �

Works because of Julia’s multiple dispatch feature
▶ Infix x * y is just syntax for *(x,y)

▶ By length(methods(*)) , there are over 230 instances of *() in Julia 1.10,
for different pairs of argument types

▶ A’ is just syntax for adjoint(A) that has about 40 pre-defined methods.
▶ Above code “overloads” adjoint for new argument type LinOp , enabling A’ .

No need for syntax like A.adjoint(), enabling reuse of matrix-like code!
17 / 36



Multiple dispatch continued (skip) J. Fessler
CI in Julia

▶ Many computational imaging system models involve products of linear operators
◦ super-resolution forward model has blur and down-sampling
◦ compressed sensing MRI has coil sensitivity, Fourier transform, sampling

▶ Consider A = BC for which Ax = B(Cx) and A′y = C ′(B′y)� �
Base.:(*)(A::LinOp, x) = A.forward(x) # for A*x
Base.:(*)(B::LinOp, C::LinOp) = # for B*C (composition)

LinOp(B.forward ◦ C.forward, C.adjoint ◦ B.adjoint)� �
▶ Multiple dispatch depends on all argument types (not just first argument)
▶ Julia compiler chooses most specific version of function
▶ The ◦ above is function composition.

Finite-difference regularizer in 2D J. Fessler
CI in Julia

2D finite difference regularizer for CI with periodic boundary conditions in sum form:

R(x) = β
N∑

n=1

M∑
m=1

ϕ(x [n, m] − x [n − 1 mod N, m]) + ϕ(x [n, m] − x [n, m − 1 mod M])

=
MN∑
k=1

ϕ([T1x]k) + ϕ([T2x]k) =
2MN∑
k=1

ϕ([Tx]k), T ≜

[
T1
T2

]
= 1′ϕ.(Tx),

where ϕ.(v) denotes applying potential function ϕ element-wise to v :

ϕ(v1)
ϕ(v2)

...

 .

If ϕ(t) = |t| then 1′ϕ.(Tx) = ∥Tx∥1 , aka anisotropic TV.

19 / 36



Finite-difference regularizer in 4 lines of Julia J. Fessler
CI in Julia� �

include("linop.jl")

diff2d_forw = x -> cat(x - circshift(x, (0,1)), x - circshift(x, (1,0)), dims=3)
diff2d_back = y -> (y[:,:,1] - circshift(y[:,:,1], (0,-1))) +

(y[:,:,2] - circshift(y[:,:,2], (-1,0)))
T = LinOp(diff2d_forw, diff2d_back)

M,N = 6,8 # testing
x = randn(M,N)
y = randn(M,N,2)
@assert sum(y .* (T * x)) ≈ sum(x .* (T' * y)) # check adjoint <T*x,y> ≈ <x,T'y>� �
▶ In math: T1 = I − P1 =⇒ T1x = x − P1x,

where P1 denotes the circular shift along first index
▶ Here the resulting linear operator T maps a 2D array into a 3D array.
▶ T’ is the adjoint (not transpose) of that operator.
▶ .* does element-wise multiplication

Regularizer gradient J. Fessler
CI in Julia

Gradient of regularizer uses derivative ϕ̇ of potential function element-wise:

R(x) =
∑

k
ϕ([Tx]k) = 1′ϕ.(Tx) =⇒ ∇R(x) = T ′ϕ̇.(Tx).

Example: Fair potential function ϕ̇(z) = β
z

1 + |z | /δ� �
δ = 0.1 # parameter for Fair potential function
β = 5 # regularization parameter
dϕ(z) = β * z / (1 + abs(z)/δ) # derivative of potential function
∇R(x) = T' * dϕ.(T * x) # regularizer gradient� �
▶ f.(v) applies f element-wise in Julia
▶ Julia matches math or vice-versa?

Image denoising example (skip) J. Fessler
CI in Julia

y = x + ε so A = I. Use x0 = y� �
include("rls.jl"); include("diff2d.jl") # makes T
using LinearAlgebra: I
using ImagePhantoms: shepp_logan, SheppLoganEmis
using MIRTjim: jim
using Random: seed!
using Plots: plot
M,N = 120,128; xtrue = shepp_logan(M,N,SheppLoganEmis()) # digital phantom
σ = 0.2; seed!(0); y = xtrue + σ * randn(size(xtrue)) # noisy data

δ = 0.1 # parameter for Fair potential function
β = 5; dφ(z) = β * z / (1 + abs(z)/δ) # derivative of potential function
∇R(x) = T' * dφ.(T * x) # regularizer gradient
α = 1 / (1 + 8β) # step size = 1 / Lipschitz constant
xh = rls(y, I, y, α, ∇R ; Niter=60)
plot(jim(y, title="Noisy", clim=(0,8)), # jiffy image display

jim(xh, title="Denoised", clim=(0,8))) # savefig("denoise-y-xh.pdf")� �

Image denoising result J. Fessler
CI in Julia

Super-resolution system model J. Fessler
CI in Julia

Simple 2×-downsampling model:
y [m, n] = x [2m, 2n] + ε[m, n] corresponds to y = Ax + ε for A : R2M×2N 7→ RM×N� �
include("linop.jl") # defines LinOp
down2_forw = (x) -> x[1:2:end,1:2:end] # down-sampling in 1 line
function down2_adjoint(y) # up-sampling (adjoint) takes 2 lines:

x = zeros(eltype(y), size(y).*2)
x[1:2:end,1:2:end] .= y # key step for up-sampling
return x

end
A = LinOp(down2_forw, down2_adjoint)

M,N = 60,64; x = randn(2M,2N); y = randn(M,N) # test data
@assert sum(y .* (A * x)) ≈ sum(x .* (A' * y)) # check adjoint� �
▶ .= does element-wise (in-place) assignment

24 / 36



Super-resolution example J. Fessler
CI in Julia� �

include("rls.jl"); include("diff2d.jl") # makes T
include("down2-linop.jl") # makes A
using ImagePhantoms: shepp_logan, SheppLoganEmis
using MIRTjim: jim
using Random: seed!
using Plots: plot, savefig
using ImageFiltering: imfilter
M,N = 120,128; xtrue = shepp_logan(M,N,SheppLoganEmis()) # digital phantom
σ = 0.1; seed!(0); y = A*xtrue; y += σ * randn(size(y)) # noisy data

x0 = imfilter(A'*y, ones(2,2)) # initial guess
δ = 0.1 # parameter for Fair potential function
β = 2; dφ(z) = β * z / (1 + abs(z)/δ) # derivative of potential function
∇R(x) = T' * dφ.(T * x) # regularizer gradient
α = 1 / (1 + 8β) # 1 / Lipschitz constant (∥A∥2 = 1)
xh = rls(y, A, x0, α, ∇R ; Niter=600) # same RLS as before!
plot(jim(y, title="Noisy Lo-Res data", clim=(0,8)),

jim(xh, title="Recovered", clim=(0,8))) # savefig("superres-y-xh.pdf")� �

Super-resolution results J. Fessler
CI in Julia

JuliaImageRecon, etc. J. Fessler
CI in Julia

▶ GitHub organization: https://github.com/JuliaImageRecon
• SPECTrecon.jl
• Sinograms.jl
• ImagePhantoms.jl
• TrainingPhantoms.jl
• NeuralImagingNetworks.jl

▶ MRI https://github.com/MagneticResonanceImaging
• BlochSim.jl
• MRIFieldmaps.jl
• MRIReco.jl
• ...

More organizations / packages J. Fessler
CI in Julia

• https://github.com/JuliaMath/FFTW.jl

• https://github.com/JuliaNLSolvers/Optim.jl

• https://github.com/FluxML

• https://github.com/JuliaDiff/ForwardDiff.jl
Ala carte style...

Critiques? J. Fessler
CI in Julia

▶ Many plotting back-ends including pyplot
◦ Interact package provides GUI elements (sliders, buttons) in Jupyter
◦ Time-to-first-plot continually improving (much better in Julia 1.10)

▶ No clear function
x = nothing frees memory used by x

▶ Julia default indexing is 1-based, not 0-based
◦ for x in list

for (i,x) in enumerate(list)
◦ Can create variables with other indexing conventions
◦ FFTViews package enables [−N/2, . . . , N/2 − 1] indexing for DSP
◦ Array comprehension largely eliminates need for ndgrid

[function(j,k,l) for j in 0:J-1, k in 0:0.5:K-1, l in 1:L]
▶ Encapsulation can be circumvented
▶ Default is 1 thread (config file option to multithread)
▶ Obsolete Q/A on stackexchange etc.

Random notes J. Fessler
CI in Julia

▶ Detailed documentation at https://docs.julialang.org
Online tutorials, stackexchange, github conversations, wikibook:
https://en.wikibooks.org/wiki/Introducing_Julia

▶ 7:0.1:100000 stored as Range type, not as a huge array
▶ Julia does not force you to indent any particular way
▶ Functions can return a tuple of variables� �

rotate = (x,y,θ) -> (x*cos(θ) + y*sin(θ), -x*sin(θ) + y*cos(θ)) # 3 in, 2 out
xnew, ynew = rotate(x, y, π/4) # use both outputs� �

▶ Some of many useful packages for CI:
JuliaImages: image processing. NFFT.jl: Nonuniform FFT. MRIReco.jl
GIRFReco.jl [3]

Markdown function comments / documentation J. Fessler
CI in Julia� �

"""
This function applies `Niter` iterations of GD
to compute the minimizer ``\\hat{x}`` of the cost function
``\\Psi(x) = \\frac{1}{2} \\| A x - y \\|_2^2 + R(x)``
given the gradient ``\\nabla R``
"""
function rls(y, A, x, α::Real, ∇R::Function ; Niter::Int = 50)

∇Ψ(x) = A'*(A*x - y) + ∇R(x) # reg. LS cost gradient
gd(∇Ψ, α, x ; Niter) # returns updated x

end� �

31 / 36



Multiple dispatch comparison J. Fessler
CI in Julia

From 2019 JuliaCon talk by Stefan Karpinski https://youtu.be/kc9HwsxE1OY?t=679

Juno/Atom in action / python integration J. Fessler
CI in Julia

• Python integration
• @

• Debugger feedback
From 2019 JuliaCon talk
by Mike Innes
https://youtu.be/OcUXjk7DFvU?t=1055

Further reading J. Fessler
CI in Julia

▶ Oct. 2019 SIAM News article about differentiable programming in Julia
▶ Forward-mode automatic differentiation [4] with ForwardDiff.jl
▶ Reverse-mode automatic differentiation [5] with Zygote.jl

▶ Compilation to cloud TPUs [6]
▶ Optimization package(s) [7]
▶ GPU integration [8]
▶ Intro to machine learning with Julia

https://tinyurl.com/ml2-18-jf
▶ Help wanted!

https://github.com/JeffFessler/MIRT.jl/blob/master/doc/matlab-to-julia.md

Acknowledgment: Beamer slides made with modified jlcode.sty from
https://github.com/wg030/jlcode

Resources J. Fessler
CI in Julia

Longer talk and code available online at
https://www.ima.umn.edu/2019-2020/SW10.14-18.19/28302
http://web.eecs.umich.edu/~fessler/papers/files/talk/19/ima-julia

